MATH590: C2examples, clustering of leukocytes

In this module homework you are invited to explore the use of clustering methods to determine different types of white blood cells

Blood image database

Read data and carry out visual examination of images

Choose directory of images to process

In[1]:=
```
SetDirectory["/home/student/courses/MATH590/SETdata/BloodCells/JPEGImages"]
```
Out[1]=
```
/home/student/courses/MATH590/SETdata/BloodCells/JPEGImages
```

List the files in the current directory

In[2]:=
```
f = FileNames[]; Nim = Length[f]
```
Out[2]=
```
366
```

In[3]:=
```
Manipulate[Import[f[[i]]], {i, 1, Nim, 1}]
```

i



Here are two typical images that differentiate between WBC and RBC by staining (WBC appear blue), and also show different morphology of WBCs.

In[4]:=
```
{im1 = Import[f[[1]]], im2 = Import[f[[9]]]}
```

Out[4]= {, }

The image data can be separated into individual color channels, adjusted to cover the entire intensity scale (contrast enhancement)

In[5]:=
```
{imR, imG, imB} = Map[ImageAdjust[#] &, ColorSeparate[im1]]
```

Out[5]=



It can also be reduced to a grayscale image, again adjusted to cover the entire intensity scale

In[6]:=
```
imBW = ImageAdjust[ColorNegate[ColorConvert[im1, "Grayscale"]]]
```

Out[6]=



Isolating leukocytes by Wright-Giemsa staining

The additional information provided by chemical staining and the relative paucity of leukocytes (i.e., WBC are well separated in the blood smear data) allows straightforward identification of white blood cells.

Repeat the isolation of the individual color channels, and adjust for maximum contrast.

In[7]:=
```
{imR1, imG1, imB1} = Map[ImageAdjust[#] &, ColorSeparate[im1]]
```

Out[7]= {  ,  ,  }

The green channel immediately furnishes a mask to eliminate the background fluid

In[8]:=
```
mask1 = ColorNegate[Binarize[imG1]]
```

Out[8]=

Eliminate the background from the blue channel data

In[9]:=
```
cellsB1 = ImageMultiply[mask1, imB1]
```

Out[9]=



Generate a mask for the leukocytes

In[10]:=
```
Lmask1 = FillingTransform[Binarize[cellsB1, 0.6]]
```

Out[10]=



Filter out small components from the mask

In[11]:=
```
MaxCells = 100;
MinPixels = Floor[Apply[Times, ImageDimensions[Lmask1]] / MaxCells];
```

In[12]:=
```
FLmask1 = Binarize[ColorConvert[
   Colorize[SelectComponents[Lmask1, #Count > MinPixels &]], "Grayscale"]]
```

Out[12]=



Define the smallest possible image that contains just leukocytes

In[13]:=
```
wbc = ImageCrop[ImageMultiply[FLmask1, im1]]
```

Out[13]=



Encapsulate all these operations into a single function

In[14]:=

```
Leukocytes[im_, t_] :=
 Module[{imR, imG, imB, mask, MaxCells, MinPixels, cellsB, Lmask, FLmask, wbc},
  {imR, imG, imB} = Map[ImageAdjust[#] &, ColorSeparate[im]];
  mask = ColorNegate[Binarize[imG]];
  MaxCells = 100;
  MinPixels = Floor[Apply[Times, ImageDimensions[mask]] / MaxCells];
  cellsB = ImageMultiply[mask, imB];
  Lmask = FillingTransform[Binarize[cellsB, t]];
  FLmask = Binarize[ColorConvert[
     Colorize[SelectComponents[Lmask, #Count > MinPixels &]], "Grayscale"]];
  wbc = ImageCrop[ImageMultiply[FLmask, im]];
  Return[wbc];
 ]
```

In[15]:=

```
Leukocytes[im2, 0.6]
```

Out[15]=



In[16]:=
```
Manipulate[Leukocytes[Import[f[[k]]], t],
 {k, 1, Length[f], 1}, {{t, 0.6}, 0.5, 0.7}]
```

Out[16]=



Homework questions

Question 1

Carefully read through the image processing carried out to remove Wright-Giemsa staining in C2examples.pdf. Identify at least five steps in which a clustering (i.e., set partitioning) technique was used. Refer to the step by the In[#] label, and comment on the type of clustering you think is being applied.

Question 2

Again with reference to the removal of Wright-Giemsa staining in C2examples.pdf, identify at least three steps in which the concepts of real analysis (perhaps in discretized form) are being used.

Question 3

Randomly choose 100 images. Isolate the leukocytes and find their size (binarize the image and count the number of ones). Identify clusters of leukocyte sizes.

Question 4

For the chosen 100 images, apply the GradientFilter operation to isolate the perimeter, and measure the perimeter by counting the number of edge pixels. Identify clusters of leukocytes by perimeter size.

Question 5

Finally define the circularity as the ratio of the equivalent disk perimeter to the measured perimeter. Identify clusters of leukocytes by circularity.